

**Příklad 5.5:** Uvažujme model tvaru:


$$Y = \sin(X_1) + 7 \cdot \sin^2(X_2) + \frac{1}{10} \cdot X_3^4 \cdot \sin(X_1),$$

kde  $X_1, X_2, X_3$  jsou všechny náhodné veličiny s rovnoměrným rozdělením pravděpodobnosti na intervalu  $(-\pi, \pi)$ . Provedme globální analýzu citlivosti.

**Řešení:** Nejprve určíme střední hodnotu a rozptyl řešení  $Y$ . To provedeme výpočtem pomocí systému Maple, viz zdrojový kód Maple na konci. Obdržíme tak:

$$E(Y) = \frac{7}{2}, \quad V(Y) = \frac{1}{1800} \cdot \pi^8 + \frac{1}{50} \cdot \pi^4 + \frac{53}{8}.$$

Pravděpodobnostní rozdělení veličiny  $Y$  můžeme odhadnout vygenerováním realizací pro  $X_1, X_2, X_3$ , následným vyhodnocením a zobrazením histogramu (v Maple příkazem `Histogram`). Histogram získaný v systému Maple pro  $M = 10^5$  vyhodnocení modelové rovnice je uveden na obrázku 5.1.



**Obr. 5.1** Histogram vygenerovaných řešení pro model příkladu 5.5.

Podle vztahu (5.8)

$$V_i = V_{X_i}(E_{X_{-i}}(Y|X_i)), \quad (5.8)$$

vypočítáme hodnoty vlivů prvního řádu jednotlivých parametrů  $X_i$  na řešení  $Y$  (pro  $i = 1, 2, 3$ ):

$$V_1 = V_{X_1}(E(Y|X_1)) = V\left(\sin(X_1) + \frac{\pi^4}{50} \cdot \sin(X_1)\right) = \frac{1}{5000} \cdot \pi^8 + \frac{1}{50} \cdot \pi^4 + \frac{1}{2},$$

$$V_2 = V_{X_2}(E(Y|X_2)) = V(-7 \cdot \cos^2(X_2)) = \frac{49}{8},$$

$$V_3 = V_{X_3}(E(Y|X_3)) = V\left(\frac{7}{2}\right) = 0.$$

Následně určíme podle vztahu (5.9)

$$S_i = \frac{V_i}{V(Y)}. \quad (5.9)$$

indexy globální citlivosti pro parametry  $X_1, X_2, X_3$ :

$$S_1 = \frac{V_1}{V(Y)} \doteq 0,3139; \quad S_2 = \frac{V_2}{V(Y)} \doteq 0,4424; \quad S_3 = \frac{V_3}{V(Y)} = 0.$$

V tomto jednoduchém příkladu bylo možné získat indexy globální citlivosti analyticky zavedenými vztahy. Pro názornost vypočítejme nyní indexy globální citlivosti pomocí Sobol'ovy metody.

Nejprve musíme vygenerovat dvě skupiny realizací parametrů  $X_1, X_2, X_3$ . Necht' každá skupina obsahuje  $M = 10^5$  realizací pro každý z parametrů. Zvolíme jednu ze skupin, spočítáme  $M$  vyhodnocení modelu  $Y$  (pro jednotlivé realizace parametrů  $X_1, X_2, X_3$ ) a určíme jejich průměr (tj. střední hodnotu) a rozptyl. Obdržíme<sup>1</sup>:

$$\widehat{E}(Y) \doteq 3,4958; \quad \widehat{V}(Y) \doteq 13,9476; \quad (V(Y) \doteq 13,8446).$$

Nyní přistoupíme k výpočtu odhadů vlivů jednotlivých parametrů definovaných vztahem (5.16) až (5.18).

$$\widehat{f}_0 = \frac{1}{M} \cdot \sum_{k=1}^M f(x_{1k}, \dots, x_{mk}), \quad (5.16)$$

$$\widehat{V}(Y) = \frac{1}{M} \cdot \sum_{k=1}^M f^2(x_{1k}, \dots, x_{mk}) - \left(\widehat{f}_0\right)^2, \quad (5.17)$$

$$\widehat{V}_i = \frac{1}{M} \cdot \sum_{k=1}^M f(x_{-ik}^{(1)}, x_{ik}^{(1)}) \cdot f(x_{-ik}^{(2)}, x_{ik}^{(1)}) - \left(\widehat{f}_0\right)^2, \quad (5.18)$$

$$\widehat{V}_{-i} = \frac{1}{M} \cdot \sum_{k=1}^M f(x_{-ik}^{(1)}, x_{ik}^{(1)}) \cdot f(x_{-ik}^{(1)}, x_{ik}^{(2)}) - \left(\widehat{f}_0\right)^2, \quad (5.19)$$

Pro výpočet vlivu  $V_1$  parametru  $X_1$  vypočteme dalších  $M$  vyhodnocení modelu  $Y$ , a to tak, že budeme brát realizace parametru  $X_1$  z první skupiny (té dříve zvolené) a realizace parametrů  $X_2, X_3$  z druhé skupiny na začátku vygenerovaných realizací. Zcela analogicky vypočteme jiných  $M$  vyhodnocení modelu  $Y$  jak pro určení vlivu  $V_2$  parametru  $X_2$ , tak pro určení vlivu  $V_3$  parametru  $X_3$ . Získáme níže uvedené hodnoty:

$$\widehat{S}_1 \doteq 0,3131; \quad \widehat{S}_2 \doteq 0,4418; \quad \widehat{S}_3 \doteq -0,0040.$$

Po výpočtu vlivů  $V_1, V_2, V_3$  můžeme přistoupit k výpočtu indexů globální citlivosti prvního řádu podle vztahu (5.9):

$$\widehat{V}_1 \doteq 4,4250; \quad \widehat{V}_2 \doteq 6,2190; \quad \widehat{V}_3 \doteq 0,0010;$$

$$\widehat{S}_1 \doteq 0,3173; \quad \widehat{S}_2 \doteq 0,4459; \quad \widehat{S}_3 \doteq 0,0001;$$

<sup>1</sup> V závorce je uvedena pro porovnání zaokrouhlená hodnota přesné hodnoty  $V(Y)$ .

Jestliže aplikujeme korekční člen (5.21),

$$\left| \frac{1}{M} \cdot \sum_{k=1}^M f(\mathbf{x}_{(k)}^{(1)}) \cdot f(\mathbf{x}_{(k)}^{(2)}) - (\widehat{f}_0)^2 \right|. \quad (5.21)$$

jehož hodnota je přibližně  $-0,0571$ . Homma a Saltelli (viz [24]) doporučují korekční člen (5.21) aplikovat bez absolutní hodnoty<sup>2</sup> na jednotlivé rozptyly  $V_{i_1, \dots, i_s}$  (pro  $i_1, \dots, i_s \in \{1, 2, 3\}$ ,  $i_1 < \dots < i_s$ ) tak, že od rozptylů prvního řádu ( $V_i$ ) se korekční člen odečte. Pak obdržíme:

$$\widehat{V}_1 \doteq 4,3680; \quad \widehat{V}_2 \doteq 6,1619; \quad \widehat{V}_3 \doteq -0,0561.$$

Základní zdrojový kód pro výpočet výše uvedených hodnot v Maple:

### Výpočet indexů globální citlivosti pomocí systému Maple

```
M := 10^5:
with(Statistics):
X[1] := RandomVariable(Uniform(-Pi, Pi)):
X[2] := RandomVariable(Uniform(-Pi, Pi)):
X[3] := RandomVariable(Uniform(-Pi, Pi)):

X1s := Sample(X[1], M): # M realizaci veliciny X[1]
X2s := Sample(X[2], M): # M realizaci veliciny X[2]
X3s := Sample(X[3], M): # M realizaci veliciny X[3]

Ys := sin~(X1s) + 7*(sin~(X2s))^~2 + ((1/10)*X3s^~4)*(sin~(X1s)):

f[0] := Mean(Ys): # odhad stredni hodnoty
V[0] := Variance(Ys): # odhad rozptylu

X1s2 := Sample(X[1], M): # "druha skupina" M realizaci veliciny X[1]
X2s2 := Sample(X[2], M): # "druha skupina" M realizaci veliciny X[2]
X3s2 := Sample(X[3], M): # "druha skupina" M realizaci veliciny X[3]
Ys2 := Vector(M):

# citlivost parametru X[1]
Ys2 := sin~(X1s) + 7*(sin~(X2s2))^~2 + ((1/10)*X3s2^~4)*~(sin~(X1s)):
V[1] := Mean(Ys*~Ys2)-f[0]^2:
S[1] = V[1]/V[0]:

# citlivost parametru X[2]
Ys3 := sin~(X1s2) + 7*(sin~(X2s))^~2 + ((1/10)*X3s2^~4)*~(sin~(X1s2)):
V[2] := Mean(Ys*~Ys3)-f[0]^2:
S[2] = V[2]/V[0]:

# citlivost parametru X[3]
Ys4 := sin~(X1s2) + 7*(sin~(X2s2))^~2 + ((1/10)*X3s^~4)*~(sin~(X1s2)):
V[3] := Mean(Ys*~Ys4)-f[0]^2:
S[3] = V[3]/V[0]:

# korekcni clen
Ysk := sin~(X1s2) + 7*(sin~(X2s2))^~2 + ((1/10)*X3s2^~4)*~(sin~(X1s2)):
```

<sup>2</sup> Absolutní hodnota byla přidána, aby příslušná věta měla správný matematický význam. Znaménko členu v absolutní hodnotě je však podstatné!

```
V[k] := Mean(Ys*~Ysk) - f[0]^2;
```